
1 'MAKIMAKI 

10 CLS:CLV:CLP 

20 LET[0],1,33,32,31,-1,-33,-32,-31 

30 H=#921:M=#A8F:V=0:W=#900 

40 POKEW,6:COPYW+1,W,30 

50 FORA=0TO22:POKE#91F+A*32,6,6:NEXT 

60 COPY#BE1,#BE0,31 

70 POKEH,#E8:POKEM,#ED 

80 PLAY"T200O4G2AG8A8B8<D2":WAIT140 

90 IFINKEY()V=(V+1)%8 

100 H=H+[V] 

110 C=PEEK(H):IFCGOTO180 

120 POKEH,#E8 

130 N=[RND(4)*2]*B 

140 IFPEEK(M+N)N=0 

150 POKEM,0:M=M+N:POKEM,#ED 

160 BEEP6*B:WAIT8:BEEP7*B:B=!B 

170 GOTO90 

180 IFC=237GOTO190ELSEGOTO260 

190 LC12,10:?"ツカマエタ!!"; 

200 PLAY"T200O4A2BAG2 

210 FORA=1TO5 

220 POKE#76A,#85,#A1:WAIT8 

230 POKE#76A,#A1,#85:WAIT8 

240 NEXT 

250 GOTO320 

260 LC12,10:?"CLASH!!"; 

270 PLAY"T200O4AA+G-2G2 

280 FORA=1TO8 

290 POKEH,#E8+F:F=!F:WAIT8 

300 NEXT 

310 POKEH,42 

320 IFINKEY()=10GOTO10ELSECONT 

 
 
■ 変数 ■ 
 
A ： 汎用（FOR～NEXT ループなど） 
B ： おもち君用（0：移動せず、音鳴らず １：移動、音が鳴る） 
C ： ヘビ君移動先のキャラクターコード（110 行のチェックで移動先が空白でない場合 180 行へ） 
F ： 「CLASH!!」時の○●点滅用 
H ： ヘビ君の VRAM アドレス 
M ： おもち君の VRAM アドレス 
N ： おもち君移動加算量（右なら 1、下なら 32 を VRAM アドレスに加算する） 
V ： ヘビ君の移動加算量の配列に入る値（0～7 順に変化） 
W ： 壁左上の VRAM アドレス 

 
 
■ 処理の内訳 ■ 
 
10～80 行 ： 初期設定（変数リセット、壁描画、キャラクター描画、OP ミュージック） 
90～120 行 ： ヘビ君のルーチン 
130～170 行 ： おもち君のルーチン 
180 行 ： ゲームオーバーの判定 
190～250 行 ： 「ツカマエタ！！」演出 
260～310 行 ： 「CLASH!!」演出 
320 行 ： 再スタート受付(ENTER キーが押されたら 10 行へ) 

 
 
★ 160 行の WAIT8 の値を変えると、全体のスピードが変更できます。 

